// File: contracts/BubbleGumGame.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import './BubbleGumJuice.sol';

contract BubbleGumGame is BubbleGumJuice {
  constructor(string memory _name, string memory _symbol, uint _launchAt) BubbleGumJuice(_name, _symbol, _launchAt) {}
}
// File: contracts/BubbleGumJuice.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./BubbleGumAction.sol";

contract BubbleGumJuice is BubbleGumAction {
  constructor(string memory _name, string memory _symbol, uint _launchAt) BubbleGumAction(_name, _symbol, _launchAt) {}

  function tokenOfStakerByIndex(address _staker, uint256 _index) public view returns (uint256) {
    require(_index < totalStakedGums[_staker], "Staker index out of bounds");
    return _stakedGums[_staker][_index];
  }

  function snappable(uint _id) public view returns (uint) {
    if (meta[_id].lastSnap == 0) return 0;

    uint delta = block.number - meta[_id].lastSnap;
    uint amt = meta[_id].size * meta[_id].intensity * delta * vars[Var.CHEW_RATE];

    return amt;
  }

  function snap(uint _id) onlyStakeOwner(_id) whenNotPaused public {
    uint amt = snappable(_id);
    uint split = amt / vars[Var.STAKE_SPLIT];
    amt -= split;

    if (amt > 0) _juicy.mint(msg.sender, amt);
    if (split > 0) _juicy.mint(address(this), split);

    meta[_id].lastSnap = block.number;

    emit Snap(amt, msg.sender);
  }

  function stake(uint _id) onlyTokenOwner(_id) whenNotPaused external {
    meta[_id].lastSnap = block.number;
    uint256 length = totalStakedGums[msg.sender];
    _stakedGums[msg.sender][length] = _id;
    _stakedGumsIndex[_id] = length;
    stakeOwners[_id] = msg.sender;
    totalStakedGums[msg.sender]++;

    _transfer(msg.sender, address(this), _id);
    emit Stake(_id, msg.sender);
  }

  function unstake(uint _id) onlyStakeOwner(_id) whenNotPaused external {
    snap(_id);

    uint256 lastIdx = --totalStakedGums[msg.sender];
    uint256 idx = _stakedGumsIndex[_id];
    if (idx != lastIdx) {
      uint256 lastId = _stakedGums[msg.sender][lastIdx];
      _stakedGums[msg.sender][idx] = lastId;
      _stakedGumsIndex[lastId] = idx;
    }
    delete _stakedGumsIndex[_id];
    delete _stakedGums[msg.sender][lastIdx];
    delete stakeOwners[_id];
    meta[_id].lastSnap = 0;

    bool isDropped = _roll("drop", _id, vars[Var.PROBA_DROP]);
    if (isDropped) {
      return _destroy(_id);
    } else {
      uint gasBurn = 0;
      for (uint i = 0; i < 250; i++) gasBurn++; // Prevent low-pass exploit.

      _transfer(address(this), msg.sender, _id);
      emit Unstake(_id, msg.sender);
    }
  }
}
// File: contracts/BubbleGumAction.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./BubbleGumStore.sol";

contract BubbleGumAction is BubbleGumStore {
  constructor(string memory _name, string memory _symbol, uint _launchAt) BubbleGumStore(_name, _symbol, _launchAt) {}

  function _removeGenesis(uint _id) private {
    uint256 lastIdx = --totalGenesis;
    uint256 idx = _genesisIdx[_id];
    if (idx != lastIdx) {
      uint256 lastId = genesis[lastIdx];
      genesis[idx] = lastId;
      _genesisIdx[lastId] = idx;
    }
    delete _genesisIdx[_id];
    delete genesis[lastIdx];
  }

  function _destroy(uint _id) internal {
    if (meta[_id].isGenesis) _removeGenesis(_id);
    delete meta[_id];
    _burn(_id);

    emit Destroy(_id, msg.sender);
  }

  function blow(uint _id) onlyTokenOwner(_id) whenNotPaused external {
    require(meta[_id].size < 25, "Max size reached.");

    _juicy.burn(msg.sender, vars[Var.FEE_BLOW]);

    bool isBurst = _roll("burst", _id, vars[Var.PROBA_BURST]);
    if (isBurst) {
      return _destroy(_id);
    } else {
      uint gasBurn = 0;
      for (uint i = 0; i < 420; i++) gasBurn++; // Prevent low-pass exploit.

      meta[_id].size++;

      bool isFren = meta[_id].isGenesis ? _roll("fren", _id, vars[Var.PROBA_FREN]) : false;
      if (isFren) {
        uint amt = _juicy.balanceOf(address(this));
        if (amt > 0) {
          _juicy.burn(address(this), amt);
          _juicy.mint(msg.sender, amt);
          emit Frens(_id, amt, msg.sender);
        }
      }
      emit Blow(_id, meta[_id].size, msg.sender);
    }
  }

  function join(uint _a, uint _b) onlyTokenOwner(_a) onlyTokenOwner(_b) whenNotPaused external {
    _juicy.burn(msg.sender, vars[Var.FEE_JOIN]);

    uint8 flavorA = meta[_a].flavor;
    uint8 flavorB = meta[_b].flavor;
    uint8 flavor = flavorA | flavorB;
    require(flavor != flavorA && flavor != flavorB, "Must create a new flavor.");
    require(meta[_a].isGenesis == meta[_b].isGenesis, "Must be same type.");

    uint size = meta[_a].size + meta[_b].size;
    uint intensity = 0;

    for (uint8 i = 0; i < 8; i++) intensity += _taste(flavor, i) ? 1 : 0;

    bool isGenesis = meta[_a].isGenesis;
    _destroy(_a);
    _destroy(_b);

    bool isDropped = _roll("join", _a, vars[Var.PROBA_DROP]);
    if (isDropped) {
      return;
    } else {
      uint id = _mintGum(isGenesis, size, flavor, intensity, msg.sender);
      emit Join(_a, _b, id);
    }
  }
}
// File: contracts/BubbleGumStore.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "@openzeppelin/contracts/utils/Counters.sol";
import './BubbleGumRoll.sol';
import './BubbleGumController.sol';

contract BubbleGumStore is BubbleGumRoll, BubbleGumController {
  using Counters for Counters.Counter;

  constructor(string memory _name, string memory _symbol, uint _launchAt) BubbleGumController(_name, _symbol, _launchAt) {}

  function _rollGenesisOwner(uint _id) private view returns (address) {
    uint rndIdx = _rnd("genesis", _id) % totalGenesis;
    uint id = genesis[rndIdx];
    address owner = ownerOf(id);

    return owner == address(this) ? stakeOwners[id] : owner;
  }

  function _mintGum(
    bool _isGenesis,
    uint _size,
    uint8 _flavor,
    uint _intensity,
    address _recipient
  ) internal returns (uint) {
    _ids.increment();
    uint id = _ids.current();

    Meta memory bubbleGum;
    bubbleGum.size = _size;
    bubbleGum.flavor = _flavor;
    bubbleGum.intensity = _intensity;
    bubbleGum.isGenesis = _isGenesis;
    meta[id] = bubbleGum;

    if (_isGenesis) {
      _genesisIdx[id] = totalGenesis;
      genesis[totalGenesis] = id;
      totalGenesis++;
    }
    _mint(_recipient, id);

    return id;
  }

  function mint(bool _isGenesis, uint _maxJuicy) whenNotPaused external payable returns (uint) {
    require(block.timestamp > launchAt, "Hasn't launched.");

    if (_isGenesis) {
      require(totalSupply() < vars[Var.TOTAL_GENESIS], "Max supply.");
      require(msg.value >= vars[Var.FEE_GENESIS], "Requires fee.");
    } else {
      uint fee = _juicy.totalSupply() * 1000 / (vars[Var.TARGET_SUPPLY] - totalSupply());
      require(_maxJuicy == 0 || fee <= _maxJuicy, "Price slipped.");
      _juicy.burn(msg.sender, fee);
    }

    uint8 flavor = uint8(2 ** (_rnd("flavor", _ids.current()) % 8));
    bool isShared = _roll("mint", _ids.current(), vars[Var.PROBA_SHARE]);
    address recipient = totalGenesis > 0 ? _rollGenesisOwner(_ids.current()) : msg.sender;

    return _mintGum(_isGenesis, 1, flavor, 1, (_isGenesis || !isShared) ? msg.sender : recipient);
  }
}
// File: @openzeppelin/contracts/utils/Counters.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @title Counters
 * @author Matt Condon (@shrugs)
 * @dev Provides counters that can only be incremented, decremented or reset. This can be used e.g. to track the number
 * of elements in a mapping, issuing ERC721 ids, or counting request ids.
 *
 * Include with `using Counters for Counters.Counter;`
 */
library Counters {
    struct Counter {
        // This variable should never be directly accessed by users of the library: interactions must be restricted to
        // the library's function. As of Solidity v0.5.2, this cannot be enforced, though there is a proposal to add
        // this feature: see https://github.com/ethereum/solidity/issues/4637
        uint256 _value; // default: 0
    }

    function current(Counter storage counter) internal view returns (uint256) {
        return counter._value;
    }

    function increment(Counter storage counter) internal {
        unchecked {
            counter._value += 1;
        }
    }

    function decrement(Counter storage counter) internal {
        uint256 value = counter._value;
        require(value > 0, "Counter: decrement overflow");
        unchecked {
            counter._value = value - 1;
        }
    }

    function reset(Counter storage counter) internal {
        counter._value = 0;
    }
}

// File: contracts/BubbleGumRoll.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

abstract contract BubbleGumRoll {
  function _rnd(string memory _k, uint256 _v) internal view returns (uint256) {
    return uint256(keccak256(abi.encode(tx.origin, blockhash(block.number-1), block.timestamp, _k, _v)));
  }

  function _roll(string memory _k, uint _id, uint _proba) internal view returns (bool) {
    return _rnd(_k, _id) % _proba == 0;
  }
}
// File: contracts/BubbleGumController.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/Pausable.sol";
import "./BubbleGumRoll.sol";
import "./BubbleGumMeta.sol";

contract BubbleGumController is Ownable, Pausable, BubbleGumMeta {
  constructor(string memory _name, string memory _symbol, uint _launchAt) BubbleGumMeta(_name, _symbol, _launchAt) {}

  modifier onlyTokenOwner(uint _id) {
    require(ownerOf(_id) == msg.sender, "Unauthorized.");
    _;
  }

  modifier onlyStakeOwner(uint _id) {
    require(stakeOwners[_id] == msg.sender, "Unauthorized.");
    _;
  }

  function adminSetJuicy(address _address) external onlyOwner { _juicy = IJuicy(_address); }

  function adminSetVar(Var _key, uint _val) external onlyOwner { vars[_key] = _val; }

  function adminWipeGenesisJuice() public onlyOwner {
    (bool success, ) = owner().call{value: address(this).balance}("");
    require(success, "Unable to remove remaining funds.");
  }
}
// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        require(!paused(), "Pausable: paused");
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        require(paused(), "Pausable: not paused");
        _;
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: contracts/BubbleGumMeta.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import './BubbleGumState.sol';
import './library/Base64.sol';

contract BubbleGumMeta is BubbleGumState {
  using Counters for Counters.Counter;
  constructor(string memory _name, string memory _symbol, uint _launchAt) BubbleGumState(_name, _symbol, _launchAt) {}

  function _taste(uint8 _flavors, uint8 _flavor) internal pure returns (bool) {
    require(_flavor < 8, "Only 8 flavors exist.");
    bool isTasted = _flavors & (1 << _flavor) == 2 ** _flavor;

    return isTasted;
  }

  function _tokenImage(uint _id) internal view returns (string memory) {
    bytes memory img = '<svg xmlns="http://www.w3.org/2000/svg" viewBox="0 0 100 100"><defs><linearGradient id="a" gradientTransform="rotate(45)"><stop offset="0" stop-color="red"/><stop offset="0.2" stop-color="#ff0"/><stop offset="0.4" stop-color="lime"/><stop offset="0.6" stop-color="aqua"/><stop offset="0.8" stop-color="blue"/><stop offset="1" stop-color="#f0f"/></linearGradient><clipPath id="b"><circle cx="50" cy="50" r="45"/></clipPath><radialGradient id="c" r="45" gradientUnits="userSpaceOnUse"><stop offset="0" stop-opacity="0"/><stop offset="0.6" stop-opacity="0"/><stop offset="0.8" stop-opacity="0.3"/><stop offset="1" stop-opacity="0.7"/></radialGradient><linearGradient id="d" gradientTransform="rotate(90)"><stop offset="0" stop-color="#fff"/><stop offset="0.5" stop-color="#fff0"/></linearGradient></defs>';
    if (meta[_id].intensity == 8) {
      // Add rainbow background for bubble gums with maximum flavor intensity.
      img = abi.encodePacked(img, '<rect width="100" height="100" style="fill:url(#a)"/>');
    }
    img = abi.encodePacked(img, '<circle cx="50" cy="50" r="45"/><g style="clip-path:url(#b)">');
    if (_taste(meta[_id].flavor, 0)) img = abi.encodePacked(img, '<rect x="-37" y="21.1" width="120" height="10.4" transform="translate(-11.8 24) rotate(-45)" style="fill:red"/>');
    if (_taste(meta[_id].flavor, 1)) img = abi.encodePacked(img, '<rect x="-22.5" y="35.6" width="120" height="10.4" transform="translate(-17.8 38.5) rotate(-45)" style="fill:#e91e63"/>');
    if (_taste(meta[_id].flavor, 2)) img = abi.encodePacked(img, '<rect x="-29.8" y="28.3" width="120" height="10.4" transform="translate(-14.8 31.2) rotate(-45)" style="fill:#4a148c"/>');
    if (_taste(meta[_id].flavor, 3)) img = abi.encodePacked(img, '<rect x="-15.2" y="42.8" width="120" height="10.4" transform="translate(-20.8 45.7) rotate(-45)" style="fill:#2196f3"/>');
    if (_taste(meta[_id].flavor, 4)) img = abi.encodePacked(img, '<rect x="-8" y="50.1" width="120" height="10.4" transform="translate(-23.8 53) rotate(-45)" style="fill:#4caf50"/>');
    if (_taste(meta[_id].flavor, 5)) img = abi.encodePacked(img, '<rect x="6.5" y="64.6" width="120" height="10.4" transform="translate(-29.9 67.5) rotate(-45)" style="fill:#ffeb3b"/>');
    if (_taste(meta[_id].flavor, 6)) img = abi.encodePacked(img, '<rect x="-0.7" y="57.4" width="120" height="10.4" transform="translate(-26.8 60.3) rotate(-45)" style="fill:#ff9800"/>');
    if (_taste(meta[_id].flavor, 7)) img = abi.encodePacked(img, '<rect x="13.8" y="71.9" width="120" height="10.4" transform="translate(-32.9 74.8) rotate(-45)" style="fill:#fff"/>');

    img = abi.encodePacked(img, '</g><path d="M95,50A43.5,43.5,0,0,1,81.9,81.9,43.9,43.9,0,0,1,50,95,43.5,43.5,0,0,1,18.1,81.9,43.5,43.5,0,0,1,5,50,43.5,43.5,0,0,1,18.1,18.1,43.9,43.9,0,0,1,50,5,44.3,44.3,0,0,1,81.9,18.1,43.9,43.9,0,0,1,95,50Z" style="fill:url(#c)"/><path d="M31.6,16.1c9.3.2,10.5,4.8,3.5,13.6S22.6,42.2,19.8,39.1s-2.3-7.3.7-13.6S27.2,15.8,31.6,16.1Z" style="fill:#fff;opacity:0.84"/><path d="M20.8,44.8c2.3.2,3,2,2.1,5.2s-2.2,4.3-3.8,3.1a4.7,4.7,0,0,1-1.7-5.2C18,45.3,19.2,44.3,20.8,44.8Z" style="fill:#fff;opacity:0.84"/><path d="M50,6.5c19.6,0,35.6,12.7,35.6,28.3S69.6,63.1,50,63.1,14.4,50.5,14.4,34.8,30.4,6.5,50,6.5Z" style="fill:url(#d)"/><text x="50%" y="50%" dominant-baseline="middle" text-anchor="middle" style="font-size:28px;font-family:sans-serif;letter-spacing:-1px;fill:#fff;stroke:#000;stroke-width:.5px">');
    img = abi.encodePacked(img, Strings.toString(meta[_id].size));
    img = abi.encodePacked(img, '</text></svg>');

    return string(abi.encodePacked("data:image/svg+xml;base64,",Base64.encode(img)));
  }

  /**
   * Time until launch.
   */
  function untilLaunch() public view returns (uint) {
    return block.timestamp < launchAt ? (launchAt - block.timestamp) : 0;
  }

  /**
   * Token URI
   * Returns token metadata with generated SVG.
   */
  function tokenURI(uint256 _id) override public view returns (string memory) {
    require(_id <= _ids.current(), "Token doesn't exist.");

    bytes memory _out = abi.encodePacked(
      '{"name":"Bubble Gum Game #',
      Strings.toString(_id),
      '","description":"',
      DESCRIPTION,
      '","image":"',
      _tokenImage(_id),
      '","attributes":[{"trait_type":"Size","value":"',
      Strings.toString(meta[_id].size),
      '"},{"trait_type":"Intensity","value":"',
      Strings.toString(meta[_id].intensity),
      '"}'
    );

    _out = abi.encodePacked(_out,
      meta[_id].isGenesis ? ',{"value":"Genesis"}' : '',
      _taste(meta[_id].flavor, 0) ? ',{"value":"Strawberry"}' : '',
      _taste(meta[_id].flavor, 1) ? ',{"value":"Watermelon"}' : '',
      _taste(meta[_id].flavor, 2) ? ',{"value":"Grape"}' : '',
      _taste(meta[_id].flavor, 3) ? ',{"value":"Blueberry"}' : ''
    );

    _out = abi.encodePacked(_out,
      _taste(meta[_id].flavor, 4) ? ',{"value":"Lime"}' : '',
      _taste(meta[_id].flavor, 5) ? ',{"value":"Pineapple"}' : '',
      _taste(meta[_id].flavor, 6) ? ',{"value":"Orange"}' : '',
      _taste(meta[_id].flavor, 7) ? ',{"value":"Mint"}' : ''
      ']}'
    );

    return string(abi.encodePacked('data:application/json;base64,', Base64.encode(_out)));
  }
}
// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/BubbleGumState.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "@openzeppelin/contracts/token/ERC721/extensions/ERC721Enumerable.sol";
import "@openzeppelin/contracts/utils/Counters.sol";
import './interface/IJuicy.sol';
import './BubbleGumStruct.sol';
import './BubbleGumEvents.sol';

contract BubbleGumState is ERC721Enumerable, BubbleGumStruct, BubbleGumEvents {
  using Counters for Counters.Counter;
  Counters.Counter internal _ids;

  uint public totalGenesis;
  mapping(uint256 => uint256) public genesis;
  mapping(uint256 => Meta) public meta;
  mapping(uint256 => address) public stakeOwners;
  mapping(Var => uint) public vars;
  mapping(address => uint256) public totalStakedGums;

  IJuicy internal _juicy;
  mapping(uint256 => uint256) internal _genesisIdx;
  mapping(address => mapping(uint256 => uint256)) internal _stakedGums;
  mapping(uint256 => uint256) internal _stakedGumsIndex;

  string constant DESCRIPTION = "The juiciest NFT game on Ethereum, where Chewers compete to blow the biggest bubble gum bubbles with the highest flavor intensity. If they're not careful, bubbles can burst and be destroyed or their gum can be dropped and lost forever. Genesis bubble gum chewers receive extra juicy gum from their frens.";
  uint launchAt;

  constructor(string memory _name, string memory _symbol, uint _launchAt) ERC721(_name, _symbol) {
    vars[Var.TOTAL_GENESIS] = 10000;
    vars[Var.TARGET_SUPPLY] = 25000;
    vars[Var.FEE_GENESIS]   = 0.3 ether;
    vars[Var.FEE_BLOW]      = 25 ether;
    vars[Var.FEE_JOIN]      = 250 ether;
    vars[Var.PROBA_BURST]   = 2;
    vars[Var.PROBA_SHARE]   = 100;
    vars[Var.PROBA_DROP]    = 2;
    vars[Var.PROBA_FREN]    = 10000;
    vars[Var.STAKE_SPLIT]   = 40;
    vars[Var.CHEW_RATE]     = 0.001 ether;

    launchAt = _launchAt;
  }
}
// File: contracts/library/Base64.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/// [MIT License]
/// @title Base64
/// @notice Provides a function for encoding some bytes in base64
/// @author Brecht Devos <brecht@loopring.org>
library Base64 {
  bytes internal constant TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

  /// @notice Encodes some bytes to the base64 representation
  function encode(bytes memory data) internal pure returns (string memory) {
    uint256 len = data.length;
    if (len == 0) return "";

    // multiply by 4/3 rounded up
    uint256 encodedLen = 4 * ((len + 2) / 3);

    // Add some extra buffer at the end
    bytes memory result = new bytes(encodedLen + 32);

    bytes memory table = TABLE;

    assembly {
      let tablePtr := add(table, 1)
      let resultPtr := add(result, 32)

      for {
        let i := 0
      } lt(i, len) {

      } {
        i := add(i, 3)
        let input := and(mload(add(data, i)), 0xffffff)

        let out := mload(add(tablePtr, and(shr(18, input), 0x3F)))
        out := shl(8, out)
        out := add(out, and(mload(add(tablePtr, and(shr(12, input), 0x3F))), 0xFF))
        out := shl(8, out)
        out := add(out, and(mload(add(tablePtr, and(shr(6, input), 0x3F))), 0xFF))
        out := shl(8, out)
        out := add(out, and(mload(add(tablePtr, and(input, 0x3F))), 0xFF))
        out := shl(224, out)

        mstore(resultPtr, out)

        resultPtr := add(resultPtr, 4)
      }

      switch mod(len, 3)
      case 1 {
        mstore(sub(resultPtr, 2), shl(240, 0x3d3d))
      }
      case 2 {
        mstore(sub(resultPtr, 1), shl(248, 0x3d))
      }

      mstore(result, encodedLen)
    }

    return string(result);
  }
}
// File: @openzeppelin/contracts/token/ERC721/extensions/ERC721Enumerable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../ERC721.sol";
import "./IERC721Enumerable.sol";

/**
 * @dev This implements an optional extension of {ERC721} defined in the EIP that adds
 * enumerability of all the token ids in the contract as well as all token ids owned by each
 * account.
 */
abstract contract ERC721Enumerable is ERC721, IERC721Enumerable {
    // Mapping from owner to list of owned token IDs
    mapping(address => mapping(uint256 => uint256)) private _ownedTokens;

    // Mapping from token ID to index of the owner tokens list
    mapping(uint256 => uint256) private _ownedTokensIndex;

    // Array with all token ids, used for enumeration
    uint256[] private _allTokens;

    // Mapping from token id to position in the allTokens array
    mapping(uint256 => uint256) private _allTokensIndex;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(IERC165, ERC721) returns (bool) {
        return interfaceId == type(IERC721Enumerable).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721Enumerable-tokenOfOwnerByIndex}.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721.balanceOf(owner), "ERC721Enumerable: owner index out of bounds");
        return _ownedTokens[owner][index];
    }

    /**
     * @dev See {IERC721Enumerable-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _allTokens.length;
    }

    /**
     * @dev See {IERC721Enumerable-tokenByIndex}.
     */
    function tokenByIndex(uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721Enumerable.totalSupply(), "ERC721Enumerable: global index out of bounds");
        return _allTokens[index];
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual override {
        super._beforeTokenTransfer(from, to, tokenId);

        if (from == address(0)) {
            _addTokenToAllTokensEnumeration(tokenId);
        } else if (from != to) {
            _removeTokenFromOwnerEnumeration(from, tokenId);
        }
        if (to == address(0)) {
            _removeTokenFromAllTokensEnumeration(tokenId);
        } else if (to != from) {
            _addTokenToOwnerEnumeration(to, tokenId);
        }
    }

    /**
     * @dev Private function to add a token to this extension's ownership-tracking data structures.
     * @param to address representing the new owner of the given token ID
     * @param tokenId uint256 ID of the token to be added to the tokens list of the given address
     */
    function _addTokenToOwnerEnumeration(address to, uint256 tokenId) private {
        uint256 length = ERC721.balanceOf(to);
        _ownedTokens[to][length] = tokenId;
        _ownedTokensIndex[tokenId] = length;
    }

    /**
     * @dev Private function to add a token to this extension's token tracking data structures.
     * @param tokenId uint256 ID of the token to be added to the tokens list
     */
    function _addTokenToAllTokensEnumeration(uint256 tokenId) private {
        _allTokensIndex[tokenId] = _allTokens.length;
        _allTokens.push(tokenId);
    }

    /**
     * @dev Private function to remove a token from this extension's ownership-tracking data structures. Note that
     * while the token is not assigned a new owner, the `_ownedTokensIndex` mapping is _not_ updated: this allows for
     * gas optimizations e.g. when performing a transfer operation (avoiding double writes).
     * This has O(1) time complexity, but alters the order of the _ownedTokens array.
     * @param from address representing the previous owner of the given token ID
     * @param tokenId uint256 ID of the token to be removed from the tokens list of the given address
     */
    function _removeTokenFromOwnerEnumeration(address from, uint256 tokenId) private {
        // To prevent a gap in from's tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = ERC721.balanceOf(from) - 1;
        uint256 tokenIndex = _ownedTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary
        if (tokenIndex != lastTokenIndex) {
            uint256 lastTokenId = _ownedTokens[from][lastTokenIndex];

            _ownedTokens[from][tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
            _ownedTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index
        }

        // This also deletes the contents at the last position of the array
        delete _ownedTokensIndex[tokenId];
        delete _ownedTokens[from][lastTokenIndex];
    }

    /**
     * @dev Private function to remove a token from this extension's token tracking data structures.
     * This has O(1) time complexity, but alters the order of the _allTokens array.
     * @param tokenId uint256 ID of the token to be removed from the tokens list
     */
    function _removeTokenFromAllTokensEnumeration(uint256 tokenId) private {
        // To prevent a gap in the tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = _allTokens.length - 1;
        uint256 tokenIndex = _allTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary. However, since this occurs so
        // rarely (when the last minted token is burnt) that we still do the swap here to avoid the gas cost of adding
        // an 'if' statement (like in _removeTokenFromOwnerEnumeration)
        uint256 lastTokenId = _allTokens[lastTokenIndex];

        _allTokens[tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
        _allTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index

        // This also deletes the contents at the last position of the array
        delete _allTokensIndex[tokenId];
        _allTokens.pop();
    }
}

// File: contracts/interface/IJuicy.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

interface IJuicy {
  function totalSupply() external view returns (uint);
  function balanceOf(address) external returns (uint256);
  function mint(address, uint) external;
  function burn(address, uint) external;
  function transferFrom(address, address, uint256) external returns (bool);
}
// File: contracts/BubbleGumStruct.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

abstract contract BubbleGumStruct {
  enum Var {
    TOTAL_GENESIS,
    TARGET_SUPPLY,
    FEE_GENESIS,
    FEE_BLOW,
    FEE_JOIN,
    PROBA_BURST,
    PROBA_SHARE,
    PROBA_DROP,
    PROBA_FREN,
    STAKE_SPLIT,
    CHEW_RATE
  }

  struct Meta {
    bool isGenesis;
    uint size;
    uint8 flavor;
    uint intensity;
    uint lastSnap;
  }
}
// File: contracts/BubbleGumEvents.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

abstract contract BubbleGumEvents {
  event Blow(
    uint indexed id,
    uint indexed size,
    address indexed chewer
  );

  event Frens(
    uint indexed id,
    uint indexed amt,
    address indexed chewer
  );

  event Destroy(
    uint indexed id,
    address indexed chewer
  );

  event Join(
    uint indexed a,
    uint indexed b,
    uint indexed id
  );

  event Stake(
    uint indexed id,
    address indexed chewer
  );

  event Unstake(
    uint indexed id,
    address indexed chewer
  );

  event Snap(
    uint indexed amount,
    address indexed chewer
  );
}
// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        require(operator != _msgSender(), "ERC721: approve to caller");

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer of token that is not own");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional enumeration extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Enumerable is IERC721 {
    /**
     * @dev Returns the total amount of tokens stored by the contract.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns a token ID owned by `owner` at a given `index` of its token list.
     * Use along with {balanceOf} to enumerate all of ``owner``'s tokens.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) external view returns (uint256 tokenId);

    /**
     * @dev Returns a token ID at a given `index` of all the tokens stored by the contract.
     * Use along with {totalSupply} to enumerate all tokens.
     */
    function tokenByIndex(uint256 index) external view returns (uint256);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}